
Chapter 1

The lpres package

This package was written by René Hartung in 2009; maintenance has been overtaken by Laurent

Bartholdi, who translated the manual to GAPDoc.

1.1 Introduction

In 1980, Grigorchuk [Gri80] gave an example of an in�nite, �nitely generated torsion group which

provided a �rst explicit counter-example to the General Burnside Problem. This counter-example is

nowadays called the Grigorchuk group and was originally de�ned as a group of transformations of the

unit interval which preserve the Lebesgue measure. Beside being a counter-example to the General

Burnside Problem, the Grigorchuk group was a �rst example of a group with an intermediate growth

function (see [Gri83]) and was used in the construction of a �nitely presented amenable group which

is not elementary amenable (see [Gri98]).

The Grigorchuk group is not �nitely presentable (see [Gri99]). However, in 1985, Igor Lysenok

(see [Lys85]) determined the following recursive presentation for the Grigorchuk group:

ha;b;c;d j a2;b2;c2;d2;bcd; [d;da]s
n

; [d;dacaca]s
n

;(n 2 N)i;

where s is the homomorphism of the free group over fa;b;c;dg which is induced by a 7! ca;b 7!
d;c 7! b, and d 7! c. Hence, the in�nitely many relators of this recursive presentation can be described

in �nite terms using powers of the endomorphism s .

In 2003, Bartholdi [Bar03] introduced the notion of an L-presentation for presentations of this

type; that is, a group presentation of the form

G=

*
S

�����Q[
[

j2F�

Rj

+
;

whereF� denotes the free monoid generated by a set of free group endomorphismsF. He proved that

various branch groups are �nitely L-presented but not �nitely presentable and that every free group in

a variety of groups satisfying �nitely many identities is �nitely L-presented (e.g. the Free Burnside-

and the Free n-Engel groups).

The lpres-package de�nes new GAP objects to work with �nitely L-presented groups. The main

part of the package is a nilpotent quotient algorithm for �nitely L-presented groups; that is, an al-

gorithm which takes as input a �nitelyL-presented group G and a positive integer c. It computes a

lpres 4

polycyclic presentation for the lower central series quotient G=gc+1(G). Therefore, a nilpotent quo-
tient algorithm can be used to determine the abelian invariants of the lower central series sections

gc(G)=gc+1(G) and the largest nilpotent quotient of G if it exists.

Our nilpotent quotient algorithm generalizes Nickel's algorithm for �nitely presented groups (see

[Nic96]) which is implemented in the NQ-package; see [Nic03]. In difference to the NQ-package,

the lpres-package is implemented in GAP only.

Since �nite L-presentations generalize �nite presentations, our algorithm also applies to �nitely

presented groups. It coincides with Nickel's algorithm in this special case.

A detailed description of our algorithm can be found in [BEH08] or in the diploma thesis [Har08].

Furthermore the lpres-package includes the algorithms of [Har10] for approximating the Schur mul-

tiplier of �nitely L-presented groups.

The lpres-package also includes the Reidemeister-Schreier algorithm from [Har12]. L-presented

groups were introduced as a tool to understand self-similar groups such as the Grigorchuk group. As

such, lpres works in close contact with the package fr. See [Har13] for more on the relationships

between L-presented groups and self-similar groups.

Finally, we note that we use the term "algorithm" somewhat loosely: many of the algorithms in

this package are in fact semi-algorithms, guaranteed to give a correct answer but not guaranteed to

terminate.



Chapter 2

An Introduction to L-presented groups

2.1 De�nitions

Let S be an alphabet, Q and R be subsets of the free group FS over this alphabet, and F be a set of

free group endomorphisms j:FS ! FS. An L-presentation is a quadruple (S;Q;F;R) and it is called

�nite if the sets S, Q, F, and R are �nite. A (�nite) L-presentation (S;Q;F;R) de�nes the (�nitely)

L-presented group

G=

*
S

�����Q[
[

j2F�

Rj

+

where F� denotes the free monoid generated by F; that is, the closure of F[fidg under composition.

The elements in Q are the �xed relators and the elements in R are the iterated relators of the

L-presentation (S;Q;F;R). An L-presentation of the form (S; /0;F;R) is an ascending L-presentation

and it is an invariant L-presentation if the normal subgroup

K =

*
Q[

[

j2F�

Rj

+FS

is j-invariant for each j 2 F; that is, if K satis�es Kj � K for each j 2 F. Note that every ascend-

ing L-presentation is invariant and for each L-presentation (S;Q;F;R) there is a unique underlying

ascending L-presentation (S; /0;F;R) which is invariant. In general it is not decidable whether or not

a given L-presentation is invariant as this would require a solution to the word-problem.

In the remainder of this manual, an L-presented group is always �nitely L-presented.

2.2 Creating an L-presented group

The construction of an L-presented group is similar to the construction of a �nitely presented group

(see Chapter (Reference: Finitely Presented Groups) of the GAP Reference manual for further

details).

2.2.1 LPresentedGroup

. LPresentedGroup(F, frels, endos, irels) (function)

5

returns theGAP object of an L-presented group with the underlying free group F , the �xed relators

frels , the set of endomorphisms endos , and the iterated relators irels . The input variables frels

and irels need to be �nite subsets of the underlying free group F and endos needs to be a �nite list

of homomorphisms F ! F .

For example, the Grigorchuk group,D
a;b;c;d

���a2;b2;c2;d2;bcd; [d;da]s
n

; [d;dacaca]s
n

;(n 2 N0)
E
;

can be constructed as follows.
Example

gap> F:=FreeGroup( "a", "b", "c", "d" );

<free group on the generators [ a, b, c, d ]>

gap> AssignGeneratorVariables( F );

#I Assigned the global variables [ a, b, c, d ]

gap> frels:=[a^2, b^2, c^2, d^2, b*c*d];;

gap> endos:=[GroupHomomorphismByImagesNC( F, F, [a, b, c, d], [c^a, d, b, c])];;

gap> irels:=[Comm( d, d^a ), Comm( d, d^(a*c*a*c*a) )];;

gap> G:=LPresentedGroup( F, frels, endos, irels );

<L-presented group on the generators [ a, b, c, d ]>

There are various examples of �nitely L-presented groups available in the library of the lpres-

package.

2.2.2 ExamplesOfLPresentations

. ExamplesOfLPresentations(n) (function)

returns some well-known examples of �nitely L-presented groups. The input of this function

needs to be a positive integer at most 10.

n=1 The Grigorchuk group on 4 generators; cf. [Gri80], [Lys85], and [Bar03, Theorem 4.6],

n=2 the Grigorchuk group on 3 generators; cf. [Gri80], [Lys85], and [Bar03, Theorem 4.6],

n=3 the lamplighter group Z=2 oZ; cf. [Bar03, Theorem 4.1],

n=4 the Brunner-Sidki-Vieira group; cf. [BSV99] and [Bar03, Theorem 4.4],

n=5 the Grigorchuk supergroup; cf. [BG02] and [Bar03, Theorem 4.6],

n=6 the Fabrykowski-Gupta group; cf. [FG85] and [BEH08],

n=7 the Gupta-Sidki group; cf. [Sid87] and [BEH08],

n=8 an index-3 subgroup of the Gupta-Sidki group,

n=9 the Basilica group; cf. [G �Z02] and [BV05],

n=10

Baumslag's �nitely generated, in�nitely related group with a trivial multiplier; cf. [Bau71].

Furthermore, every free group in a variety of groups satisfying �nitely many identities is �nitely L-

presented. Some of these groups are available from the lpres-package using the following operations;

for further details we refer to the diploma thesis [Har08].

2.2.3 FreeEngelGroup

. FreeEngelGroup(n, num) (operation)

returns an L-presentation for the free n -Engel group on num generators; that is, the free group in

the variety of num -generated groups satisfying the n -Engel identity.

2.2.4 FreeBurnsideGroup

. FreeBurnsideGroup(exp, num) (operation)

returns an L-presentation for the free Burnside group on num generators with exponent exp ; that

is, the free group in the variety of num -generated groups with exponent exp .

2.2.5 FreeNilpotentGroup

. FreeNilpotentGroup(c, num) (operation)

returns an L-presentation for the free nilpotent group of class c on num generators; that is, the free

group in the variety of num -generated, nilpotent groups with nilpotency class c .

2.2.6 GeneralizedFabrykowskiGuptaLpGroup

. GeneralizedFabrykowskiGuptaLpGroup(n) (operation)

returns an L-presentation for the n -th generalized Fabrykowski-Gupta group as constructed in

[BEH08].

2.2.7 LamplighterGroup (llint)

. LamplighterGroup(filter, int) (operation)

. LamplighterGroup(filter, pcgroup) (operation)

returns a �nite L-presentation for the lamplighter group on int lamp states in the �rst case, if

filter is the �lter IsLpGroup. In the second case, the group pcgroup must be a �nite cyclic group.

Then the method returns a �nite L-presentation for the lamplighter group on Size(pcgroup) lamp

states; for details on the L-presentation see [Bar03].
Example

gap> LamplighterGroup( IsLpGroup, 2 );

<L-presented group on the generators [ a, t, u ]>

gap> LamplighterGroup( IsLpGroup, CyclicGroup(3) );

<L-presented group on the generators [ a, t, u ]>

2.2.8 EmbeddingOfIASubgroup

. EmbeddingOfIASubgroup(a) (operation)

computes an L-presentation for the IA-automorphism group of a free group. This is the subgroup

of automorphisms of a free group f that act trivially on the abelianization of f .

The L-presentation is taken from [DP].

Example
gap> f := FreeGroup(3);

<free group on the generators [ f1, f2, f3 ]>

gap> a := AutomorphismGroup(f);

<group of size infinity with 3 generators>

gap> ia := Source(EmbeddingOfIASubgroup(a));

<invariant LpGroup on the generators [ C(1,2), C(1,3), C(2,1), C(2,3), C(3,1), C(3,2), M(1,[2,3]),

M(2,[1,3]), M(3,[1,2]) ]>

gap> rank := 3;

3

gap> q := NilpotentQuotient(ia,rank);;

gap> lcs := LowerCentralSeries(q);;

gap> for i in [1..Length(lcs)-1] do

> r := AbelianInvariants(lcs[i]/lcs[i+1]);

> Print(i); if i>3 then Print("th"); else Print(ELM_LIST(["st","nd","rd"],i)); fi;

> Print(" quotient: abelian invariants ",r," (collected ",Collected(r),")\n");

> od;

1st quotient: abelian invariants [ 0, 0, 0, 0, 0, 0, 0, 0, 0 ] (collected [ [ 0, 9 ] ])

2nd quotient: abelian invariants [ 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0

] (collected [ [ 0, 18 ] ])

3rd quotient: abelian invariants [ 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0,

0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 2, 2, 2, 2, 2, 2, 2, 2, 2, 2,

2, 2, 2, 2, 3, 3, 3, 3, 3, 3, 3, 3, 3 ] (collected [ [ 0, 43 ], [ 2, 14 ], [ 3, 9 ] ])

2.3 The underlying free group

An L-presented group is de�ned as an image of its underlying free group. Note that these are two

different GAP objects. The elements of the L-presented group are represented by words in the under-

lying free group.

2.3.1 FreeGroupOfLpGroup

. FreeGroupOfLpGroup(lpgroup) (attribute)

Returns: the underlying free group of the L-presented group lpgroup

2.3.2 FreeGeneratorsOfLpGroup

. FreeGeneratorsOfLpGroup(lpgroup) (attribute)

Returns: the generators of the free group which underlies the L-presented group lpgroup

2.3.3 GeneratorsOfGroup

. GeneratorsOfGroup(lpgroup) (attribute)

Returns: the generators of the L-presented group lpgroup . These are the images of the genera-

tors of the underlying free group under the natural homomorphism.

2.3.4 UnderlyingElement

. UnderlyingElement(elm) (operation)

returns the preimage of an L-presented group element elm in the underlying free group. More

precisely, each element of an L-presented group is represented by an element in the free group. This

method returns the corresponding element in the free group.

2.3.5 ElementOfLpGroup

. ElementOfLpGroup(fam, elm) (operation)

returns the element in the L-presented group represented by the word elm on the generators of the

underlying free group, if fam is the family of L-presented group elements.

Example
gap> F:=FreeGroup( 2 );;

gap> G:=LPresentedGroup( F, [ F.1^2 ], [ IdentityMapping( F ) ], [ F.2 ] );;

gap> FreeGroupOfLpGroup( G ) = F;

true

gap> GeneratorsOfGroup( G );

[ f1, f2 ]

gap> FreeGeneratorsOfLpGroup( G );

[ f1, f2 ]

gap> last = last2;

false

gap> UnderlyingElement( G.1 );

f1

gap> last in F;

true

gap> ElementOfLpGroup( ElementsFamily( FamilyObj( G ) ), last2 ) in G;

true

2.4 Accessing an L-presentation

The �xed relators, the iterated relators, and the endomorphisms of an L-presented group are accessible

with the following methods.

2.4.1 FixedRelatorsOfLpGroup

. FixedRelatorsOfLpGroup(lpgroup) (attribute)

Returns: the �xed relators of the L-presented group lpgroup as elements of the underlying free

group.

2.4.2 IteratedRelatorsOfLpGroup

. IteratedRelatorsOfLpGroup(lpgroup) (attribute)

Returns: the iterated relators of the L-presented group lpgroup as elements of the underlying

free group.

2.4.3 EndomorphismsOfLpGroup

. EndomorphismsOfLpGroup(lpgroup) (attribute)

Returns: the endomorphisms of the L-presented group lpgroup as endomorphisms of the un-

derlying free group.

Example
gap> F:=FreeGroup( 2 );;

gap> G:=LPresentedGroup( F, [ F.1^2 ], [ IdentityMapping( F ) ], [ F.2 ] );

<L-presented group on the generators [ f1, f2 ]>

gap> FixedRelatorsOfLpGroup( G );

[ f1^2 ]

gap> IteratedRelatorsOfLpGroup( G );

[ f2 ]

gap> EndomorphismsOfLpGroup( G );

[ IdentityMapping( <free group on the generators [ f1, f2 ]> ) ]

2.5 Attributes and properties of L-presented groups

For the method-selection of the nilpotent quotient algorithm, an L-presented group may have the

following attributes and properties.

2.5.1 UnderlyingAscendingLPresentation

. UnderlyingAscendingLPresentation(lpgroup) (attribute)

returns the underlying ascending L-presentation of lpgroup ; that is, if lpgroup is �nitely L-

presented by (S;Q;F;R), the underlying ascending L-presentation is (S; /0;F;R).

2.5.2 UnderlyingInvariantLPresentation

. UnderlyingInvariantLPresentation(lpgroup) (attribute)

attempts to compute a �good� underlying invariant L-presentation for lpgroup ; that is, if

lpgroup is �nitely L-presented by (S;Q;F;R), then this method seeks to �nd a subset Q0 � Q such

that (S;Q0;F;R) is an invariant L-presentation. Note that there is always the underlying ascending L-

presentation (S; /0;F;R). However, for the ef�ciency of the nilpotent quotient algorithm it is important

that the subset Q0 is as big as possible.

Since it is undecidable, in general, whether or not a given L-presentation is invariant, there is no

algorithm which can determine the best possible underlying invariant L-presentation. The method

implemented for this attribute tries to compute a �good� invariant L-presentation and will return the

underlying ascending L-presentation in the worst case.

This attribute can be set manually using SetUnderlyingInvariantLPresentation. For in-

stance, the Grigorchuk groupD
a;b;c;d

���a2;b2;c2;d2;bcd; [d;da]s
n

; [d;dacaca]s
n

;(n 2 N0)
E
;

is invariantly L-presented and therefore, it should be constructed as follows:

Example
gap> F:=FreeGroup( "a", "b", "c", "d" );;

gap> AssignGeneratorVariables( F );

#I Assigned the global variables [ a, b, c, d ]

gap> frels:=[ a^2, b^2, c^2, d^2, b*c*d ];;

gap> endos:=[ GroupHomomorphismByImagesNC( F, F, [ a, b, c, d ], [ c^a, d, b, c ]) ];;

gap> irels:=[ Comm( d, d^a ), Comm( d, d^(a*c*a*c*a) ) ];;

gap> G:=LPresentedGroup( F, frels, endos, irels );

<L-presented group on the generators [ a, b, c, d ]>

gap> SetUnderlyingInvariantLPresentation( G, G );;

2.5.3 IsAscendingLPresentation

. IsAscendingLPresentation(lpgroup) (property)

checks whether the L-presentation of lpgroup is ascending; that is, if the set of �xed relators is

empty. This property is set automatically when creating an L-presented group with no �xed relators

using the function LPresentedGroup (2.2.1).

2.5.4 IsInvariantLPresentation

. IsInvariantLPresentation(lpgroup) (property)

attempts to check whether the L-presentation of lpgroup is invariant. In general, one cannot

decide whether or not a given L-presentation is invariant. There are mainly two methods implemented

for this property. The �rst method seeks to �nd a �good� underlying invariant L-presentation using

the operation UnderlyingInvariantLPresentation (2.5.2). If this latter L-presentation coincides

with the L-presentation of lpgroup , then lpgroup is invariantly L-presented. If this method fails,

then the second method uses the nilpotent quotient algorithm for L-presented groups which yields a

necessary condition for an L-presented group to be invariantly L-presented. Note that the latter method

may not terminate. For instance, both methods fail on Baumslag's �nitely generated, in�nitely related

group with trivial multiplier returned by ExamplesOfLPresentations (2.2.2).

2.5.5 EmbeddingOfAscendingSubgroup

. EmbeddingOfAscendingSubgroup(lpgroup) (attribute)

stores an embedding of an ascendingly L-presented subgroup of the L-presented group lpgroup .

This attribute is set for ascending L-presentations only. In this case, the identity map of lpgroup is

returned. This attribute is used in the FR-package which can construct various �nitely L-presented

groups. The embedding is useful for a nilpotent quotient algorithm of a non-invariantly L-presented

group.

2.6 Methods for L-presented groups

Some operations are natural extensions of the operations for �nitely generated groups. For example,

MappedWord(x,gens,imgs), when applied to a word x in an L-presented group, returns the group

element obtained by replacing each occurrence of a generator in gens by the corresponding element

in the list imgs. The lists gens and imgs need to have the same length.

Equality test of elements of L-presented groups is implemented using the operation

NqEpimorphismNilpotentQuotient (nq: NqEpimorphismNilpotentQuotient) to compare the

images in a nilpotent quotient of the group. The implemented method successively increases the

class of the considered quotient until the images differ. Hence, this method may not terminate and it

will only determine whether the elements are different.

2.6.1 EpimorphismFromFpGroup

. EpimorphismFromFpGroup(lpgroup, n) (operation)

returns an epimorphism from a �nitely presented group onto lpgroup . The �nitely presented

group is obtained from lpgroup by applying only words of length at most n in the endomorphisms

of lpgroup to the iterated relators of lpgroup .

2.6.2 SplitExtensionByAutomorphismsLpGroup

. SplitExtensionByAutomorphismsLpGroup(lpgroup, H, auts) (operation)

returns an L-presentation for the split extension of lpgroup by an L-presented or by a �nitely

presented group H . The action of a generator of H on lpgroup is given by an automorphism in the

list auts . Thus for each generator of H there must be an automorphism in the list auts .
Example

gap> F := FreeGroup( "a" );

<free group on the generators [ a ]>

gap> H := F / [ F.1^3 ];

<fp group on the generators [ a ]>

gap> U := ExamplesOfLPresentations( 8 );

<L-presented group on the generators [ t, u, v ]>

gap> aut:=GroupHomomorphismByImagesNC( U, U, [ U.1, U.2, U.3 ], [ U.2, U.3, U.1 ] );

[ t, u, v ] -> [ u, v, t ]

gap> SplitExtensionByAutomorphismsLpGroup( U, H, [ aut ] );

<L-presented group on the generators [ t, u, v, a ]>

2.6.3 AsLpGroup

. AsLpGroup(G) (operation)

returns an ascending L-presentation for a �nitely presented group G or for a free group G .

2.6.4 IsomorphismLpGroup

. IsomorphismLpGroup(G) (operation)

returns an isomorphism from a �nitely presented group G or from a free group G to the L-presented

group obtained from the method AsLpGroup (2.6.3).

Example
gap> F:=FreeGroup( 2 );

<free group on the generators [ f1, f2 ]>

gap> G:=F/[ F.1^2, F.2^2, Comm( F.1, F.2 ) ];

<fp group on the generators [ f1, f2 ]>

gap> IsomorphismLpGroup( G );

[ f1, f2 ] -> [ f1, f2 ]

gap> Range(last);

<L-presented group on the generators [ f1, f2 ]>

gap> Display(last);

generators = [ f1, f2 ]

fixed relators = [ ]

endomorphism = [

IdentityMapping( <free group on the generators [ f1, f2 ]> ) ]

iterated relators = [

f1^2,

f2^2,

f1^-1*f2^-1*f1*f2 ]



Chapter 3

Nilpotent Quotients of L-presented

groups

Our nilpotent quotient algorithm for �nitely L-presented groups generalizes Nickel's algorithm for

�nitely presented groups; see [Nic96]. It determines a nilpotent presentation for the lower central

series quotient of an invariantly L-presented group. A nilpotent presentation is a polycyclic presen-

tation whose polycyclic series re�nes the lower central series of the group (see the description in the

NQ-package for further details). In general, our algorithm determines a polycyclic presentation for

the nilpotent quotient of an arbitrary �nitely L-presented group. For further details on our algorithm

we refer to [BEH08] or to the diploma thesis [Har08].

3.1 New methods for L-presented groups

3.1.1 NilpotentQuotient

. NilpotentQuotient(g[, c]) (operation)

returns a polycyclic presentation for the class-c quotient g=gc+1(g) of the L-presented group g if

c is speci�ed. If c is not given, this method attempts to compute the largest nilpotent quotient of g

and will terminate only if g has a largest nilpotent quotient.

The following example computes the class-5 quotient of the Grigorchuk group.

Example
gap> G := ExamplesOfLPresentations( 1 );;

gap> H := NilpotentQuotient( G, 5 );

Pcp-group with orders [ 2, 2, 2, 2, 2, 2, 2, 2, 2, 2 ]

gap> lcs := LowerCentralSeries( H );

[ Pcp-group with orders [ 2, 2, 2, 2, 2, 2, 2, 2, 2, 2 ],

Pcp-group with orders [ 2, 2, 2, 2, 2, 2, 2 ],

Pcp-group with orders [ 2, 2, 2, 2, 2 ], Pcp-group with orders [ 2, 2, 2 ],

Pcp-group with orders [ 2, 2 ], Pcp-group with orders [ ] ]

gap> List( [ 1..5 ], x -> lcs[ x ] / lcs[ x+1 ] );

[ Pcp-group with orders [ 2, 2, 2 ], Pcp-group with orders [ 2, 2 ],

Pcp-group with orders [ 2, 2 ], Pcp-group with orders [ 2 ],

Pcp-group with orders [ 2, 2 ] ]

14

3.1.2 LargestNilpotentQuotient

. LargestNilpotentQuotient(g) (operation)

returns the largest nilpotent quotient of the L-presented group g if it exists. It uses the method

NilpotentQuotient (3.1.1). If g has no largest nilpotent quotient, this method will not terminate.

3.1.3 NqEpimorphismNilpotentQuotient

. NqEpimorphismNilpotentQuotient(g[, p/c]) (operation)

This method returns an epimorphism from the L-presented group g onto a nilpotent quotient. If the

optional argument is an integer c , the epimorphism is onto the maximal class-c quotient g==gc+1(g).
If no second argument is given, this method attempts to compute an epimorphism onto the largest

nilpotent quotient of g . If g does not have a largest nilpotent quotient, this method will not terminate.

If a pcp-group p is given as additional parameter, then p has to be a nilpotent quotient of g . The

method computes an epimorphism from the L-presented group g onto p .

The following example computes an epimorphism from the Grigorchuk group onto its class-5,

class-7, and class-10 quotients.

Example
gap> G := ExamplesOfLPresentations( 1 );

<L-presented group on the generators [ a, b, c, d ]>

gap> epi := NqEpimorphismNilpotentQuotient( G, 5 );

[ a, b, c, d ] -> [ g1, g2*g3, g2, g3 ]

gap> H := Image( epi );

Pcp-group with orders [ 2, 2, 2, 2, 2, 2, 2, 2, 2, 2 ]

gap> NilpotencyClassOfGroup( H );

5

gap> H := NilpotentQuotient( G, 7 );

Pcp-group with orders [ 2, 2, 2, 2, 2, 2, 2, 2, 2, 2, 2, 2, 2 ]

gap> NilpotentQuotient( G, 10 );

Pcp-group with orders [ 2, 2, 2, 2, 2, 2, 2, 2, 2, 2, 2, 2, 2, 2, 2, 2, 2, 2 ]

gap> NqEpimorphismNilpotentQuotient( G, H );

[ a, b, c, d ] -> [ g1, g2*g3, g2, g3 ]

gap> Image( last );

Pcp-group with orders [ 2, 2, 2, 2, 2, 2, 2, 2, 2, 2, 2, 2, 2 ]

3.1.4 AbelianInvariants

. AbelianInvariants(g) (operation)

computes the abelian invariants of the L-presented group g . It uses the operation

NilpotentQuotient (3.1.1).

Example
gap> G := ExamplesOfLPresentations( 1 );;

gap> AbelianInvariants( G );

[ 2, 2, 2 ]

3.2 A brief description of the algorithm

In the following we give a brief description of the nilpotent quotient algorithm for an arbitrary �nitely

L-presented group. For further details, we refer to [BEH08] and the diploma thesis [Har08].

Let (S;Q;F;R) be a �nite L-presentation de�ning the L-presented group G and let (S;Q0;F;R)
be an underlying invariant L-presentation. Write Ḡ for the invariantly L-presented group de�ned by

(S;Q0;F;R).
The �rst step in computing a polycyclic presentation for G=gc+1(G) is to determine a nilpotent

presentation for Ḡ=gc+1(Ḡ). This will be done by induction on c. The induction step of our algorithm
generalizes the induction step of Nickel's algorithm which mainly relies on Hermite normal form

computations. In order to use this rather fast linear algebra, we must require the group to be invariantly

L-presented. Therefore, the �xed relators must be handled separately by reducing to an underlying

invariant L-presentation �rst.

The induction step of our algorithm then returns a nilpotent presentation H for the quotient

Ḡ=gc+1(Ḡ) and an epimorphism d : Ḡ! H. Both are used to determine a polycyclic presentation for

the nilpotent quotient G=gc+1(G) using an extension d 0:FS ! H of the epimorphism d . The quotient

G=gc+1(G) is isomorphic to the factor group H=hQd 0

iH . We use the Polycyclic-package to compute

a polycyclic presentation for H=hQd 0

iH .
The ef�ciency of this general approach depends on the underlying invariant L-presentation

(S;Q0;F;R). The set of �xed relators Q0 should be as large as possible. Otherwise, the nilpotent

quotient H can be large even if the nilpotent quotient G=gc+1(G) is rather small.

The following example demonstrates the different behavior of our nilpotent quotient algorithm for

the Grigorchuk group with its �nite L-presentation�
fa;c;b;dg;fa2;b2;c2;d2;bcdg;fsg;f[d;da]; [d;dacaca]g

�
:

This latter L-presentation is obviously an invariant L-presentation. Hence, we can either use the prop-

erty IsInvariantLPresentation (2.5.4) or the attribute UnderlyingInvariantLPresentation

(2.5.2). First, one has to construct the group as described in Section 2.2:
Example

gap> F := FreeGroup( "a", "b", "c", "d" );

<free group on the generators [ a, b, c, d ]>

gap> AssignGeneratorVariables( F );

#I Assigned the global variables [ a, b, c, d ]

gap> rels := [ a^2, b^2, c^2, d^2, b*d*c ];;

gap> endos := [ GroupHomomorphismByImagesNC( F, F, [ a, b, c, d ], [ c^a, d, b, c ]) ];;

gap> itrels := [ Comm( d, d^a ), Comm( d, d^(a*c*a*c*a) ) ];;

gap> G := LPresentedGroup( F, rels, endos, itrels );

<L-presented group on the generators [ a, b, c, d ]>

gap> List( rels, x -> x^endos[1] );

[ a^-1*c^2*a, d^2, b^2, c^2, d*c*b ]

The property IsInvariantLPresentation (2.5.4) can be set manually using

SetInvariantLPresentation:
Example

gap> SetIsInvariantLPresentation( G, true );

gap> NilpotentQuotient( G, 4 );

Pcp-group with orders [ 2, 2, 2, 2, 2, 2, 2, 2 ]

gap> StringTime( time );

" 0:00:00.032"

On the other hand, one can use the attribute UnderlyingInvariantLPresentation (2.5.2) as fol-

lows:
Example

gap> U := LPresentedGroup( F, rels, endos, itrels );

<L-presented group on the generators [ a, b, c, d ]>

gap> SetUnderlyingInvariantLPresentation( G, U );

gap> NilpotentQuotient( G, 4 );

Pcp-group with orders [ 2, 2, 2, 2, 2, 2, 2, 2 ]

gap> StringTime( time );

" 0:00:00.028"

For saving memory the �rst method should be preferred in this case. In general, the L-presentation is

not invariant (or not known to be invariant) and thus the underlying invariant L-presentation has fewer

�xed relators than the group G itself. In this case, the second method is the method of choice.

There is a brute-force method implemented for the operation

UnderlyingInvariantLPresentation (2.5.2) which works quite well on the

ExamplesOfLPresentations (2.2.2). However, in the worst case, this method will return the

underlying ascending L-presentation. The following example shows the in�uence of this choice to

the runtime of the nilpotent quotient algorithm. After de�ning the group G as above, we set the

attribute UnderlyingInvariantLPresentation (2.5.2) as follows:

Example
gap> SetUnderlyingInvariantLPresentation( G, UnderlyingAscendingLPresentation(G) );

gap> NilpotentQuotient( G, 4 );

Pcp-group with orders [ 2, 2, 2, 2, 2, 2, 2, 2 ]

gap> StringTime( time );

" 0:00:02.700"

3.3 Nilpotent Quotient Systems for invariant L-presentations

For an invariantly L-presented group G, our algorithm computes a nilpotent presentation for

G=gc+1(G) by computing a weighted nilpotent quotient system for G=G0 and extending it

inductively to a weighted nilpotent quotient system for G=gc+1(G).
In the lpres package, a weighted nilpotent quotient system is a record containing the following

entries:

Lpres

the invariantly L-presented group G.

Pccol

FromTheLeftCollector (polycyclic: FromTheLeftCollector) of the nilpotent quotient rep-

resented by this quotient system.

Imgs

the images of the generators of the L-presented group G under the epimorphism onto the nilpo-

tent quotient Pccol . For each generator of G there is an integer or a generator exponent list. If

the image is an integer int , the image is a de�nition of the int -th generator of the nilpotent

presentation Pccol .

Epimorphism

an epimorphism from the L-presented group G onto its nilpotent quotient Pccol with the im-

ages of the generators given by Imgs .

Weights

a list of the weight of each generator of the nilpotent presentation Pccol .

De�nitions

the de�nition of each generator of Pccol . Each generator in the quotient system has a de�nition

as an image or as a commutator of the form [a j;ai] where a j and ai are generators of a certain

weight. If the i -th entry is an integer, the i -th generator of Pccol has a de�nition as an image.

Otherwise, the i -th entry is a 2-tuple [k; l] and the i -th generator has a de�nition as commutator

[ak;al].

A weighted nilpotent quotient system of an invariantly L-presented group can be computed with the

following functions.

3.3.1 InitQuotientSystem

. InitQuotientSystem(lpgroup) (operation)

computes a weighted nilpotent quotient system for the abelian quotient of the L-presented group

lpgroup .

3.3.2 ExtendQuotientSystem

. ExtendQuotientSystem(QS) (operation)

extends the weighted nilpotent quotient system QS for a class-c quotient of an invariantly L-

presented group to a weighted nilpotent quotient system of its class-c+1 quotient.

Example
gap> G := ExamplesOfLPresentations( 1 );

<L-presented group on the generators [ a, b, c, d ]>

gap> Q := InitQuotientSystem( G );

rec( Lpres := <L-presented group on the generators [ a, b, c, d ]>,

Pccol := <<from the left collector with 3 generators>>,

Imgs := [ 1, [ 2, 1, 3, 1 ], 2, 3 ], Epimorphism := [ a, b, c, d ] ->

[ g1, g2*g3, g2, g3 ], Weights := [ 1, 1, 1 ], Definitions := [ 1, 3, 4 ]

)

gap> ExtendQuotientSystem( Q );

rec( Lpres := <L-presented group on the generators [ a, b, c, d ]>,

Pccol := <<from the left collector with 5 generators>>,

Imgs := [ 1, [ 2, 1, 3, 1 ], 2, 3 ],

Definitions := [ 1, 3, 4, [ 2, 1 ], [ 3, 1 ] ],

Weights := [ 1, 1, 1, 2, 2 ], Epimorphism := [ a, b, c, d ] ->

[ g1, g2*g3, g2, g3 ] )

3.4 Attributes of L-presented groups related with the nilpotent quotient

algorithm

To avoid repeated extensions of a weighted nilpotent quotient system the largest known quotient sys-

tem is stored as an attribute of the invariantly L-presented group. For non-invariantly L-presented

groups (or groups which are not known to be invariantly L-presented) the known epimorphisms onto

the nilpotent quotients are stored as an attribute.

3.4.1 NilpotentQuotientSystem

. NilpotentQuotientSystem(lpgroup) (attribute)

stores the largest known weighted nilpotent quotient system of an invariantly L-presented group.

Example
gap> G := ExamplesOfLPresentations( 1 );;

gap> NilpotentQuotient( G, 5 );

Pcp-group with orders [ 2, 2, 2, 2, 2, 2, 2, 2, 2, 2 ]

gap> NilpotentQuotientSystem( G );

rec( Lpres := <L-presented group on the generators [ a, b, c, d ]>,

Pccol := <<from the left collector with 10 generators>>,

Imgs := [ 1, [ 2, 1, 3, 1 ], 2, 3 ],

Definitions := [ 1, 3, 4, [ 2, 1 ], [ 3, 1 ], [ 4, 2 ], [ 4, 3 ], [ 7, 1 ],

[ 8, 2 ], [ 8, 3 ] ], Weights := [ 1, 1, 1, 2, 2, 3, 3, 4, 5, 5 ],

Epimorphism := [ a, b, c, d ] -> [ g1, g2*g3, g2, g3 ] )

gap> NilpotencyClassOfGroup( PcpGroupByCollectorNC( last.Pccol ) );

5

3.4.2 NilpotentQuotients

. NilpotentQuotients(lpgroup) (attribute)

stores all known epimorphisms onto the nilpotent quotients of lpgroup . The nilpotent quotients

are accessible by the operation Range (Reference: Range of a general mapping).

Example
gap> G:=ExamplesOfLPresentations( 3 );;

gap> HasIsInvariantLPresentation( G );

false

gap> NilpotentQuotient( G, 3 );

Pcp-group with orders [ 0, 2, 2, 2 ]

gap> NilpotentQuotients( G );

[ [ a, t, u ] -> [ g2, g1, g2 ], [ a, t, u ] -> [ g2, g1, g2 ],

[ a, t, u ] -> [ g2, g1, g2 ] ]

gap> Range( last[2] );

Pcp-group with orders [ 0, 2, 2 ]

The underlying invariant L-presentation has stored its largest weighted nilpotent quotient system as

an attribute.
Example

gap> NilpotentQuotientSystem( UnderlyingInvariantLPresentation( G ) );

rec( Lpres := <L-presented group on the generators [ a, t, u ]>,

Pccol := <<from the left collector with 9 generators>>, Imgs := [ 1, 2, 3 ],

Definitions := [ 1, 2, 3, [ 2, 1 ], [ 3, 2 ], [ 4, 1 ], [ 4, 2 ], [ 5, 2 ],

[ 5, 3 ] ], Weights := [ 1, 1, 1, 2, 2, 3, 3, 3, 3 ],

Epimorphism := [ a, t, u ] -> [ g1, g2, g3 ] )

3.5 The Info-Class InfoLPRES

To get some information about the progress of the algorithm, one can use the info class InfoLPRES

(3.5.1).

3.5.1 InfoLPRES

. InfoLPRES (info class)

is the info class of the lpres-package. If the info-level is 1, the info-class gives further information

on the progress of the nilpotent quotient algorithm for L-presented groups. The info-level 2 also

includes some information on the runtime of our algorithm while the info-level 3 is mainly used for

debugging-purposes. An example of such a session for the Grigorchuk group is shown below:

Example
gap> SetInfoLevel( InfoLPRES, 1 );;

gap> G:=ExamplesOfLPresentations( 1 );

#I The Grigorchuk group on 4 generators

<L-presented group on the generators [ a, b, c, d ]>

gap> NilpotentQuotient( G, 3 );

#I Class 1: 3 generators with relative orders: [ 2, 2, 2 ]

#I Class 2: 2 generators with relative orders: [ 2, 2 ]

#I Class 3: 2 generators with relative orders: [ 2, 2 ]

Pcp-group with orders [ 2, 2, 2, 2, 2, 2, 2 ]

gap> SetInfoLevel( InfoLPRES, 2 );

gap> NilpotentQuotient( G, 5 );

#I Time spent for spinning algo: 0:00:00.004

#I Class 4: 1 generators with relative orders: [ 2 ]

#I Runtime for this step 0:00:00.028

#I Time spent for spinning algo: 0:00:00.008

#I Class 5: 2 generators with relative orders: [ 2, 2 ]

#I Runtime for this step 0:00:00.036

Pcp-group with orders [ 2, 2, 2, 2, 2, 2, 2, 2, 2, 2 ]

3.5.2 InfoLPRES_MAX_GENS

. InfoLPRES_MAX_GENS (global variable)

this global variable sets the limit of generators whose relative order will be shown on each step of

the nilpotent quotient algorithm, if the info-level of InfoLPRES (3.5.1) is positive.

Subgroups of L-presented groups

As shown in [Har11] it is possible to deal with �nite index subgroups of L-presented groups algorith-

mically. The lpres-package provides straightforward methods to deal with these subgroups.

The Reidemeister-Schreier algorithm from [Har12] is implemented, and computes presentations

for such subgroups.

4.1 Creating a subgroup of an L-presented group

There are two ways of de�ning subgroups of �nite index of an lpgroup . The �rst is to de�ne the

subgroup by its generators while the second de�nes the subgroup by a coset-table. Generators of

subgroup of the latter type can be computed with the usual Schreier-algorithm.

4.1.1 Subgroup

. Subgroup(G, gens) (function)

creates the subgroup U of G generated by gens . The Parent value of U will be G .

For example, the branching subgroup of the Grigorchuk group can be de�ned as follows, and a

presentation can be computed using IsomorphismLpGroup (2.6.4):

Example
gap> G := ExamplesOfLPresentations(1);;

gap> a := G.1;; b := G.2;; c := G.3;; d := G.4;;

gap> K := Subgroup( G, [ Comm( a, b ), Comm( b^a, d ), Comm( b, d^a ) ] );

Group([ a^-1*b^-1*a*b, b^-1*a^-1*d^-1*a*b*a^-1*d*a, a^-1*b^-1*a*d^-1*a^-1*b*a*d ])

gap> iso := IsomorphismLpGroup(K);

[ a^-1*b^-1*a*b, a^-1*b^-1*a*d^-1*a^-1*b*a*d, b^-1*a^-1*d^-1*a*b*a^-1*d*a ] ->

[ x1^-1*x13^-1*x12*x3, x1^-1*x13^-1*x12*x8^-1*x22^-1*x23*x24*x11, x3^-1*x12^-1*x18^-1*x29*x21*x1 ]

gap> Range(iso);

<LpGroup with 98 generators>

4.1.2 SubgroupLpGroupByCosetTable

. SubgroupLpGroupByCosetTable(G, Tab) (operation)

creates the subgroup U of G which is represented by the coset-table Tab .

21

For instance, the branching subgroup of the Grigorchuk group can be de�ned by the following

coset-table:
Example

gap> Tab := [ [ 2, 1, 6, 9, 10, 3, 11, 12, 4, 5, 7, 8, 15, 16, 13, 14 ],

[ 2, 1, 6, 9, 10, 3, 11, 12, 4, 5, 7, 8, 15, 16, 13, 14 ],

[ 3, 6, 1, 5, 4, 2, 8, 7, 10, 9, 12, 11, 14, 13, 16, 15 ],

[ 3, 6, 1, 5, 4, 2, 8, 7, 10, 9, 12, 11, 14, 13, 16, 15 ],

[ 4, 7, 5, 1, 3, 8, 2, 6, 13, 14, 15, 16, 9, 10, 11, 12 ],

[ 4, 7, 5, 1, 3, 8, 2, 6, 13, 14, 15, 16, 9, 10, 11, 12 ],

[ 5, 8, 4, 3, 1, 7, 6, 2, 14, 13, 16, 15, 10, 9, 12, 11 ],

[ 5, 8, 4, 3, 1, 7, 6, 2, 14, 13, 16, 15, 10, 9, 12, 11 ] ]

gap> U := SubgroupLpGroupByCosetTable( G, Tab );

Group(<A>subgroup of L-presented group, no generators known</A>)

gap> U = K;

true

The generators of U can be computed with the Schreier-algorithm which is implemented in the method

GeneratorsOfGroup (Reference: GeneratorsOfGroup).

4.2 Computing the index of �nite-index subgroups

In principle, it is possible to compute the index of a �nite index subgroup of an lpgroup [Har11].

The method reduces the case to certain �nitely presented groups by applying only �nitely many en-

domorphisms to the iterated relations. It then uses coset enumeration for �nitely presented groups to

compute an upper bound on the index of the subgroup. If the coset enumeration for �nitely presented

groups terminated, the method attempts to prove that the upper bound is sharp. For further details we

refer to [Har11].

4.2.1 IndexInWholeGroup

. IndexInWholeGroup(H) (method)

. FactorCosetAction(G, H) (method)

The �rst command attempts to compute the index of H in its parent group. The second one gives

the permutation action of G on the right cosets of H .
Example

gap> G := ExamplesOfLPresentations(1);;

gap> a := G.1;; b := G.2;; c := G.3;; d := G.4;;

gap> K := Subgroup( G, [ Comm(a,b), Comm( b, d^a ), Comm( b^a, d )] );;

gap> IndexInWholeGroup( K );

16

gap> FactorCosetAction( G, K );

[ a, b, c, d ] -> [ (1,2)(3,6)(4,9)(5,10)(7,11)(8,12)(13,15)(14,16),

(1,3)(2,6)(4,5)(7,8)(9,10)(11,12)(13,14)(15,16), (1,4)(2,7)(3,5)(6,8)(9,13)(10,14)(11,15)(12,16),

(1,5)(2,8)(3,4)(6,7)(9,14)(10,13)(11,16)(12,15) ]

4.2.2 Index

. Index(H, I) (method)

attempts to compute the index of I in the subgroup H . The subgroup I must be contained in H .
Example

gap> G := ExamplesOfLPresentations(1);;

gap> a := G.1;; b := G.2;; c := G.3;; d := G.4;;

gap> K := Subgroup( G, [ Comm(a,b), Comm( b, d^a ), Comm( b^a, d )] );;

gap> KxK := Subgroup( G, [ Comm(b,d^a), Comm(b^a,d), Comm(d^a,c^(a*c)),

</A> Comm( d^(a*c), c^a), Comm( d, c^(a*c*a) ), Comm( d^(a*c*a), c) ] );;

gap> Index( K, KxK );

4

4.2.3 CosetTableInWholeGroup

. CosetTableInWholeGroup(H) (method)

computes a coset-table for the subgroup H in its parent group.

Example
gap> CosetTableInWholeGroup( K );

[ [ 2, 1, 6, 9, 10, 3, 11, 12, 4, 5, 7, 8, 15, 16, 13, 14 ],

[ 2, 1, 6, 9, 10, 3, 11, 12, 4, 5, 7, 8, 15, 16, 13, 14 ],

[ 3, 6, 1, 5, 4, 2, 8, 7, 10, 9, 12, 11, 14, 13, 16, 15 ],

[ 3, 6, 1, 5, 4, 2, 8, 7, 10, 9, 12, 11, 14, 13, 16, 15 ],

[ 4, 7, 5, 1, 3, 8, 2, 6, 13, 14, 15, 16, 9, 10, 11, 12 ],

[ 4, 7, 5, 1, 3, 8, 2, 6, 13, 14, 15, 16, 9, 10, 11, 12 ],

[ 5, 8, 4, 3, 1, 7, 6, 2, 14, 13, 16, 15, 10, 9, 12, 11 ],

[ 5, 8, 4, 3, 1, 7, 6, 2, 14, 13, 16, 15, 10, 9, 12, 11 ] ]

4.3 Technical details

For performance issues the following global variables can be used to modify the behaviour of the

coset enumeration:

4.3.1 LPRES_TCSTART

. LPRES_TCSTART (global variable)

de�nes the maximal word-length of endomorphisms in the free monoid which are applied to the

iterated relations.

4.3.2 LPRES_CosetEnumerator

. LPRES_CosetEnumerator (global variable)

de�nes the coset enumeration process used for �nitely presented groups. It should be a function

which take as input a subgroup h of a �nitely presented group and it computes a coset table in the

whole group. The default uses the following method of the ACE-package
Example

function ( h )

local f, rels, gens;

f := FreeGeneratorsOfFpGroup( Parent( h ) );

rels := RelatorsOfFpGroup( Parent( h ) );

gens := List( GeneratorsOfGroup( h ), UnderlyingElement );

return ACECosetTable( f, rels, gens : silent := true,

hard := true,

max := 10 ^ 8,

Wo := 10 ^ 8 );

If the ACE-package is not available, the library coset enumeration process is used.

Approximating the Schur multiplier

The algorithm in [Har10] approximates the Schur multiplier of an invariantly �nitely L-presented

group by the quotients in its Dwyer-�ltration. This is implemented in the lpres-package and the

following methods are available:

5.1 Methods

5.1.1 GeneratingSetOfMultiplier

. GeneratingSetOfMultiplier(lpgroup) (operation)

uses Tietze transformations for computing an equivalent set of relators for lpgroup so that a

generating set for its Schur multiplier can be read off easily.

5.1.2 FiniteRankSchurMultiplier

. FiniteRankSchurMultiplier(lpgroup, c) (operation)

computes a �nitely generated quotient of the Schur multiplier of lpgroup . The method computes

the image of the Schur multiplier of lpgroup in the Schur multiplier of its class-c quotient.

5.1.3 EndomorphismsOfFRSchurMultiplier

. EndomorphismsOfFRSchurMultiplier(lpgroup, c) (operation)

computes a list of endomorphisms of the `FiniteRankSchurMultiplier' of lpgroup . These are the

endomorphisms of the invariant L-presentation induced to `FiniteRankSchurMultiplier'.

5.1.4 EpimorphismCoveringGroups

. EpimorphismCoveringGroups(lpgroup, d, c) (operation)

computes an epimorphism of the covering group of the class-d quotient onto the covering group

of the class-c quotient.

25



lpres 26

5.1.5 EpimorphismFiniteRankSchurMultiplier

. EpimorphismFiniteRankSchurMultiplier(lpgroup, d, c) (operation)

computes an epimorphism of the d-th `FiniteRankSchurMultiplier' of the invariant lpgroup onto

the c-th `FiniteRankSchurMultiplier'. Its restricts the epimorphism `EpimorphismCoveringGroups'

to the corresponding �nite rank multipliers.

5.1.6 ImageInFiniteRankSchurMultiplier

. ImageInFiniteRankSchurMultiplier(lpgroup, c, elm) (function)

computes the image of the free group element elm in the c -th `FiniteRankSchurMultiplier'. Note

that elm must be a relator contained in the Schur multiplier of lpgroup ; otherwise, the function fails

in computing the image.

The following example tackels the Schur multiplier of the Grigorchuk group.

Example
gap> G := ExamplesOfLPresentations( 1 );;

gap> gens := GeneratingSetOfMultiplier( G );

rec( FixedGens := [ b^-2*c^-2*d^-2*b*c*d*b*c*d ],

IteratedGens := [ d^-1*a^-1*d^-1*a*d*a^-1*d*a,

d^-1*a^-1*c^-1*a^-1*c^-1*a^-1*d^-1*a*c*a*c*a*d*a^-1*c^-1*a^-1*c^-1*a^

-1*d*a*c*a*c*a ],

BasisGens := [ a^2, b*c*d, b^-2*d^-2*b*c*d*b*c*d, b^-2*c^-2*b*c*d*b*c*d ],

Endomorphisms := [ [ a, b, c, d ] -> [ a^-1*c*a, d, b, c ] ] )

gap> H := FiniteRankSchurMultiplier( G, 5 );

Pcp-group with orders [ 2, 2, 2 ]

gap> GeneratorsOfGroup( H );

[ g15, g17, g16 ]

gap> EndomorphismsOfFRSchurMultiplier( G, 5 );

[ [ g15, g16, g17 ] -> [ g15, id, g16 ] ]

gap> Kernel( last[1] );

Pcp-group with orders [ 2 ]

gap> GeneratorsOfGroup( last );

[ g16 ]

gap> EpimorphismFiniteRankSchurMultipliers( G, 5, 2 );

[ g15, g16, g17 ] -> [ g10, id, g13 ]

gap> Range( last ) = FiniteRankSchurMultiplier( G, 2 );

true

gap> Kernel( EpimorphismFiniteRankSchurMultipliers( G, 5, 2 ) );

Pcp-group with orders [ 2 ]

gap> GeneratorsOfGroup( last );

[ g16 ]

gap> Kernel( EpimorphismFiniteRankSchurMultipliers( G, 5, 2 ) ) =

</A> Kernel( EndomorphismsOfFRSchurMultiplier( G, 5 )[1] );

true

gap> ImageInFiniteRankSchurMultiplier( G, 5, gens.FixedGens[1] );

g15

gap> ImageInFiniteRankSchurMultiplier(G,5,Image(gens.Endomorphisms[1],

</A> gens.IteratedGens[1] ) );

g16

gap> ImageInFiniteRankSchurMultiplier(G,5,gens.IteratedGens[1] );

g17



Chapter 6

On a parallel nilpotent quotient

algorithm

We included a parallel version of lpres's nilpotent quotient algorithm using the ParGAP-package

of GAP; see [Coo04]. In this chapter, we outline the basic usage of this parallel part of the lpres-

package. For further details on the parallel GAP-sessions we refer to the ParGAP-manual [Coo04].

We note that the ParGAP-package has some bottlenecks in practice. Nevertheless the signi�cant

speed-up of our computations on a multiple-core system shows that this is a reasonable extension of

the lpres-package.

6.1 Usage

For using the parallel version of the nilpotent quotient algorithm, you will need to install the ParGAP-

package as described in its manual [Coo04]. When using Version 1.1.2 of the ParGAP-package, you

will need to apply the following patch to `pargap/lib/masslave.g' as otherwise the ParGAP-session

may crash. On a linux machine you can simply use `patch < ../../nql/gap/pargap/patch' from within

the directory `pargap/lib/'.

--- masslave.g 2001-11-16 13:17:04.000000000 +0100

+++ masslave.g 2009-05-06 12:20:19.000000000 +0200

@@ -467,8 +467,9 @@

if Length(deltas)>1 then max2 := Maximum(max2, deltas[Length(deltas)-1]); fi;

max1 := deltas[Length(deltas)];

pos1 := Position( List(slaveArray, x->realtime-x.time), max1 );

- if max1 > slaveTaskTimeFactor and max1 > 30

- and slaveTaskTime[pos2].total > 60 then

+ if max1 > slaveTaskTimeFactor and

+ max1 > 30 and pos2 <> fail and

+ slaveTaskTime[pos2].total > 60 then

Print("SLAVE ",pos1," SEEMS DEAD!!\n");

fi;

end);

Now, you are ready for creating a ParGAP-session and you can load the lpres-package from within

ParGAP using `RequirePackage' as usual. The same methods as described previously are available.

28

The following example shows the application of the `NilpotentQuotient'-method to the Grigorchuk

group on a quad-core machine. Note that the signi�cant speed-up of the nilpotent quotient algorithm

is especially noticeable for large nilpotent quotients. This parallel version of lpres successfully com-

putes some nilpotent quotients which normally took more than a month to complete.
Example

GAP4, Version: 4.4.12 of 17-Dec-2008, i686-pc-linux-gnu-gcc

GAP4, Version: 4.4.12 of 17-Dec-2008, i686-pc-linux-gnu-gcc

GAP4, Version: 4.4.12 of 17-Dec-2008, i686-pc-linux-gnu-gcc

GAP4, Version: 4.4.12 of 17-Dec-2008, i686-pc-linux-gnu-gcc

GAP4, Version: 4.4.12 of 17-Dec-2008, i686-pc-linux-gnu-gcc

gap> TOPCnumSlaves;

4

gap> RequirePackage("LPRES");

true

gap> G:=ExamplesOfLPresentations(1);

<L-presented group on the generators [ a, b, c, d ]>

gap> SetInfoLevel(InfoLPRES,1);

gap> NilpotentQuotient(G,2);

#I Class 1: 3 generators with relative orders: [ 2, 2, 2 ]

#I Computing a polycyclic presentation for the covering group...

#I Checking the consistency relations...

master -> 1: (AGGLOM_TASK): [ [ -3, 1 ], [ -3, 2 ], [ -2, 1 ], [ 2, -1 ],

[ 3, -1 ] ]

master -> 2: (AGGLOM_TASK): [ [ 3, -2 ], [ 1 ], [ 2 ], [ 3 ] ]

1 -> master: [ [ 0, 0, 0, 0, 0, -2, 0 ], [ 0, 0, 0, 0, 0, 0, -2 ],

[ 0, 0, 0, 0, -2, 0, 0 ], [ 0, 0, 0, 0, -2, 0, 0 ],

[ 0, 0, 0, 0, 0, -2, 0 ] ]

2 -> master: [ [ 0, 0, 0, 0, 0, 0, -2 ], [ 0, 0, 0, 0, 0, 0, 0 ],

[ 0, 0, 0, 0, 0, 0, 0 ], [ 0, 0, 0, 0, 0, 0, 0 ] ]

#I Broadcasting the slaves...

#I Inducing the endomorphisms...

master -> 1: 1

master -> 2: 2

master -> 3: 3

master -> 4: 4

3 -> master: [ 2, 1 ]

UPDATE: [ 3, [ 2, 1 ] ]

1 -> master: [ 2, 1, 8, 1 ]

UPDATE: [ 1, [ 2, 1, 8, 1 ] ]

2 -> master: [ 2, 1, 3, 1, 4, 1 ]

UPDATE: [ 2, [ 2, 1, 3, 1, 4, 1 ] ]

master -> 1: 5

master -> 2: 6

master -> 3: 7

4 -> master: [ 4, -1, 6, -1, 10, -1 ]

UPDATE: [ 4, [ 4, -1, 6, -1, 10, -1 ] ]

1 -> master: [ 6, 1, 8, 2 ]

UPDATE: [ 5, [ 6, 1, 8, 2 ] ]

2 -> master: [ 4, 2, 6, 1, 7, 1, 10, 1 ]

UPDATE: [ 6, [ 4, 2, 6, 1, 7, 1, 10, 1 ] ]

3 -> master: [ 6, 1 ]

UPDATE: [ 7, [ 6, 1 ] ]

master -> 1: 8

master -> 2: 9

master -> 3: 10

1 -> master: [ 10, 1 ]

UPDATE: [ 8, [ 10, 1 ] ]

2 -> master: [ ]

UPDATE: [ 9, [ ] ]

3 -> master: [ 10, -1 ]

UPDATE: [ 10, [ 10, -1 ] ]

#I Broadcasting the slaves...

#I Mapping the relations...

master -> 1: 1

master -> 2: 2

master -> 3: 3

master -> 4: 4

1 -> master: [ 0, 0, 0, 0, 1, 0, 0, 0, 0, 0 ]

2 -> master: [ 0, 0, 0, 2, 0, 1, 1, 0, 0, 1 ]

3 -> master: [ 0, 0, 0, 0, 0, 1, 0, 0, 0, 0 ]

4 -> master: [ 0, 0, 0, 0, 0, 0, 1, 0, 0, 0 ]

master -> 1: 5

master -> 2: 6

master -> 3: 7

1 -> master: [ 0, 0, 0, 1, 0, 1, 1, 0, 0, 1 ]

2 -> master: [ 0, 0, 0, 0, 0, 0, 0, 0, 2, 0 ]

3 -> master: [ 0, 0, 0, 0, 0, 0, 0, 4, 6, 4 ]

#I Start spinning...

#I Extend the quotient system...

#I Class 2: 2 generators with relative orders: [ 2, 2 ]

Pcp-group with orders [ 2, 2, 2, 2, 2 ]

Note that the only difference in the parallel version of the lpres-package is a parallel version of

the operation `ExtendQuotientSystem'. This latter operation covers the induction step of the nilpotent

quotient algorithm.

